

Cominciamo a parlare di contenitori estendendo qualche concetto sull'albero degli oggetti e vediamo il seguente codice sorgente:

```
Include "Parser";
Include "Verblib";
Include "Replace";

Object stanza_1 "Stanza"
with description "Ti trovi in una stanza.",
has light;

! ### inizio parte nuova ###
Object -> scatola "scatolina"
with name 'scatola' 'scatolina',
before [;
Examine: "Una piccola scatola.";
], ! E' del tutto identico a description "Una piccola scatola.",
has openable container;

Object teschio "teschio" scatola
with name 'teschio' 'cranio',
description "Chiss@`a di chi sar@`a stato?";

Object -> gabbia "gabbia di vetro"
with name 'gabbia',
article "una",
description "Una ampia gabbia di vetro.",
has enterable openable transparent container female;

Object sedia "sedia" stanza_1
with name 'sedia',
description "Una normalissima sedia di legno.",
has female supporter enterable;

! ### fine parte nuova ###

[ Initialise;
location = stanza_1;
];

Include "ItalianG"; ! Sempre alla fine del listato
```

Cominciamo dall'oggetto scatola. Vediamo che, rispetto allo stesso oggetto dell'articolo precedente abbiamo cambiato l'attributo *open* in *openable*. *Open* significa che l'oggetto è aperto, mentre *openable* significa che l'oggetto può essere aperto (e chiuso). Un oggetto che abbia l'attributo *open* ma non *openable* rimarrà permanentemente aperto e non potrà essere chiuso. Invece un oggetto che avrà l'attributo *openable* potrà venire aperto e chiuso e, di default, sarà impostato come chiuso. Provare per credere: alla prima descrizione della stanza non ci viene detto cosa c'è all'interno del forziere (giustamente, essendo chiuso), mentre se aggiungiamo *open* ci verrà indicato subito il suo contenuto.

Nota: il contrario di *open* non è *closed*, ma *~open*: vedremo in seguito che tutti gli attributi predefiniti avranno come proprio contrario il proprio nome preceduto dalla tilde (per ottenerla, basta tenere premuto il tasto Alt e digitare sulla tastiera numerica in sequenza 1 2 6).

Passiamo alla gabbia di vetro. Vediamo che, essendo verosimilmente un contenitore, le abbiamo aggiunto gli attributi *openable* e *container* (anche in questo caso, non avendo aggiunto l'attributo *open*, la gabbia di default sarà chiusa). Ma la differenza tra una scatola e una gabbia è che nella prima non ci si può entrare, nella seconda sì. Ecco a cosa serve l'attributo *enterable*: ad indicare che un giocatore può entrare in un oggetto (non necessariamente un contenitore, come vedremo per la sedia).

Vediamo anche che la gabbia di vetro ha l'attributo *transparent*, il che significa che, se il giocatore dovesse entrare nella gabbia, chiuderla e poi richiedere una descrizione della stanza in cui si trova digitando il comando *>guarda*, gli verrebbe data la descrizione della stanza (come è giusto aspettarsi, essendo appunto il vetro trasparente). Se ci fossimo dimenticati di questo attributo il giocatore potrebbe incappare in un risultato doppiamente disastroso: primo, perché la gabbia non ha l'attributo *light*, e quindi il giocatore (al suo interno) dopo averla chiusa si ritroverebbe nell'oscurità (ma ha senso una situazione del genere in una gabbia trasparente posta in una stanza illuminata?), secondo perché anche se avessimo aggiunto l'attributo *light* al giocatore non verrebbe data la descrizione della stanza (il che è ancora una situazione inverosimile per l'esempio).

In base a quel che è stato detto, quindi, non dovrebbe stupirvi che se avessimo aggiunto l'attributo *transparent* alla scatola, pur essendo chiusa di default, avremmo ottenuto anche un indizio su ciò che contiene.

Passiamo infine all'oggetto sedia. Notiamo che (come per il teschio) abbiamo una variazione nella sua dichiarazione: probabilmente vi sareste aspettati

Object -> sedia "sedia"

ma in realtà è del tutto equivalente a

Object sedia "sedia" stanza_1

ossia la freccia tra *Object* e il nome dell'oggetto può essere (eliminata e) sostituita dal nome dell'oggetto all'interno del quale si trova, posto a fine dichiarazione. Per gusti personali, pur essendo forse più confusionario, io preferisco creare l'albero degli oggetti tramite frecce, ma sono solo, appunto, gusti personali, per questo ho indicato questo nuovo modo solamente ora.

Vediamo che la sedia ha un nuovo attributo: *supporter*, il che significa che su di essa possono essere messi oggetti. Non solo: essendo una sedia su di essa ci si può sedere, e per far ciò, purtroppo poco intuitivamente, bisogna lo stesso aggiungere, oltre a *supporter*, l'attributo *enterable*. Vedremo come definire meglio questa situazione quando parleremo di come creare e modificare i verbi: se infatti durante il gioco si digita *> siediti* ci verrà chiesto *Su cosa vuoi sederti?* e, purtroppo, si potrà rispondere anche *> gabbia*, col risultato che entreremo nella gabbia (tutto questo a causa di un'ambiguità dei verbi *siediti* ed *enter*). Per il momento non pensateci, che comunque non è un "errore grave".

Comunque, avendo definito la sedia come un supporto, saremo liberi di prendere il teschio e metterlo sopra di essa. Naturalmente non sopra il forziere o sopra la gabbia, visto che sono privi dell'attributo *supporter*.

Anche per questa volta è tutto. Vi lascio con due screenshot: cercate di compilare un codice sorgente che dia le stesse risposte

```
| Stanza del teletrasporto                                Punti: 0      Azioni: 5
Stanza del teletrasporto
Sei nella sala del teletrasporto. Scotty è già qui che aspetta che tu entri nella colonna di luce per fiondarti sul pianeta Vulcan.

Puoi vedere una colonna di luce (che è vuota) e una grossa scatola (nella quale c'è una scatola media (che è chiusa)) qui.

>apri scatola media
Hai aperto la scatola media, trovando una scatola piccola.

>apri scatola piccola
Hai aperto la scatola piccola, trovando un tricorder.

>esamina tricorder
Indispensabile per ogni missione di sbarco.

>entra in colonna
Ti trovi dentro la colonna di luce.

>guarda

Stanza del teletrasporto (dentro la colonna di luce)
Sei nella sala del teletrasporto. Scotty è già qui che aspetta che tu entri nella colonna di luce per fiondarti sul pianeta Vulcan.

Puoi vedere una grossa scatola (nella quale c'è una scatola media (nella quale c'è una scatola piccola (nella quale c'è un tricorder))) qui.
```

```
| Oscurità                                                Punti: 0      Azioni: 4
Versione 1 -- Numero di serie 150210
Inform v6.30 -- Libreria 6/11 -- Infit v2.5  S

camera
Ti trovi nella camera di una casa in cui sei entrato per rubare. Purtroppo il proprietario è già tornato: rimetti a posto il portafoglio sul comodino e un posto in cui nasconderti!

Puoi vedere un comodino e uno sgabuzzino (che è chiuso) qui.

>metti portafoglio su comodino
Hai messo il portafoglio sopra il comodino.

>apri sgabuzzino
Ora hai aperto lo sgabuzzino.

>entra in sgabuzzino
Ti trovi dentro lo sgabuzzino.

>chiudi sgabuzzino
Hai chiuso lo sgabuzzino.

È completamente buio qui!
```

[Nota: per il secondo esercizio create l'oggetto portafoglio come se fosse una stanza, poi nella routine *Initialise*, dopo la definizione della locazione iniziale, scrivete *move portafoglio to player;*]